Arduino

EZ Switch Library User Guide

A Library Supporting the Reading of Multiple Mixed-type Simple Switches & Circuits

Author: R D Bentley, Stafford, UK.

Date: March 2021

Version 1.03

Warranties & Exceptions

This document and its content are in the public domain and may be used without restriction and without warranty.

Change Record

Doc Ref/Version	Date	Change
1.02	March 2021	Initial version published
1.03	March 2021	Addition of new library variable (last switched id),
		plus example of its use in sketch with multiple buttons all
		linked to a single interrupt handler.

RDB Stafford,UK -2 - v1.03, March 2021

ARDUINO - Switch Library User Guide

Contents

Warranties & Exceptions	2
Change Record	2
Introduction	4
Overview	4
Design Objectives	5
Constraints & Limitations	5
Switch Types Supported	6
Common Switch Wiring Schemes	7
Using the <ez_switch_lib> Library</ez_switch_lib>	8
To Know	8
Location of the <ez_switch_lib> Library</ez_switch_lib>	9
Steps to Successful Use	10
Specifications	13
Specifications – Switch Control Structure (SCS)	13
Specifications - Reserved Macro Definitions and Other Declarations	14
Other Declarations	
Specifications - Switch Control Functions	
add_switch	
link_switch_to_output	
num_free_switch_slots	
read_switch	
<pre>read_button_switch read toggle switch</pre>	
print_switch	
print switches	
set debounce	
Example Sketches	
Example 1.1 - Turning LED On/Off With a Button Switch, Directly Coded	
Example 1.2 - Turning LED On/Off With a Button Switch, Indirectly Coded	
Example 2.1 - Turning LED On/Off With a Toggle Switch, Directly Coded	
Example 2.2 - Turning LED On/Off With a Toggle Switch, Indirectly Coded	28
Example 3.1 - Turning Multiple LEDs On/Off With Multiple Button & Toggle Switches, Directly	
Coded	30
Example 3.2 - Turning Multiple LEDs On/Off With Multiple Button & Toggle Switches, Indirectly	
Coded	
Example 4 – Processing More Button & Toggle Switches	
Example 5 – Using the Libraries Switch Structure Variables	
Corollary	
Switch Mismatching	
Buttons as a Toggles Many Switches, One Interrupt Service Routine (ISR)	
iviariy Switches, One interrupt Service Routine (ISR)	40

Introduction

Implementing switches, of any type, can be troublesome as not all switches are equal! Some are 'fleeting' or momentary, like button switches, and some are simply either on or off until they are 'flipped' at their next actuation. Button switches are fairly standard in their design, but toggle type switches are many varied – simple toggle, slide, tilt, rotary, etc. If you are incorporating switches into your projects then issues such as switch design, transition 'noise' and wiring schemes will all come into play at some point in a project's design.

The good news is that both types of switch can be brought to heel by the $<ez_switch_lib>$ library which provides a simple to use, no frills, software solution for connecting a mix of switch types wired in a variety of circuit schemes. The end result is that, by using the $<ez_switch_lib>$, the only components required are switches, connecting wires and, if wished, 10k ohm pull down resistors. However, even the 10k ohm resistors can be left out by choice of the right circuit (see below, Common Switch Wiring Schemes).

This User Guide (UG) describes the <code><ez_switch_lib></code> library for Arduino, detailing the functions and definitions available to the end user for implementing switches of either style and in a choice of wiring schemes - any number of switches of any style and of varying common wiring designs may be configured, the only limitation being the number of digital pins available.

However, before continuing, if you would like to understand a little about the issues associated with switches have a look at the tutorial <u>Understanding & Using Button Switches</u>. Although it is centred on the simple button switch, the basics are also common to toggle switches.

Overview

This UG provides information and guidance that will prove helpful in understanding the capabilities of the switch library, <ez_switch_lib>, in designing and implementing projects using switches, single or multiple of varying types.

The UG gives information and explanations of:

projects.

•	Design objectives	what was sought.
---	-------------------	------------------

•	Constraints & limitations	it is vitally important to understand any constraints and limitations that the library imposes/suffers, as these may play a part in the way in which you utilise the library.
•	Types of switch supported	there are many types of switch available. Those suitable for use with the library are highlighted.
•	Common wiring schemes	again there are many ways in which a switch may be wired. The library has been designed to support the two of the most common wiring schemes to be used for any switch type. The approach is to minimise any additional hardware components used.
•	Using the library	describes how the library should be (can be) incorporated into your

•	Declarations & definitions	provides a list of all of the library's switch macro definitions and control struct(ure) available to the end user to incorporate into their sketches.
•	Function specifications	each of the library's functions is detailed with an example in its use. Any specific points of note are also provided.
•	Example sketches	example sketches are provided, building from single button and toggle switches to multiple switch types using both circuit schemes and using different programming techniques to declare and use switch data.

Hopefully, the UG will become a 'one-stop-shop' to help the end user supplement understanding in the application of switches and the library's capabilities.

Design Objectives

At the outset a number of key objectives were established for the design of the switch library, these being a library that provided/supported:

- a simple, logical and straight forward design
- ease of end user project switch configuration, irrespective of type and number of switches or how connected (wired)
- different switch types the ubiquitous button switch and a variety of different types of toggle switch
- support for common wiring schemes simply connected with or without a 10k ohm pull down resistor
- mixed switch/circuit implementations support for a mix of switch types and wiring schemes
- software auto-debounce of noisy switch transitions removing from end user design consideration issues relating to noisy switch transition by incorporating transparent debounce features
- one switch read function irrespective of switch type or wiring scheme providing a simple to use function to read any and all switches
- allowing the optional linking of switches to a digital output such that auto-switching of the output can occur without end user programmatic coding
- developing a user guide that is informative and such that it is easy to 'dip' into.

Constraints & Limitations

Nothing in this world is perfect and <ez_switch_lib> is far from that. Whilst it does provide a set of useful capabilities to aid and assist Arduino project developers involving switches, there are several constraints and limitations in its design and use to be aware of:

1. Every switch to be configured requires its own digital pin. Whilst this is not an issue for say, a mega 2560 microcontroller, lesser boards are more constraining in the number of digital pins they support. Certainly for UNO microcontrollers and better, there should not be a practical issue in mapping switches to digital pins for most switch hungry projects.

- 2. Development of the library was limited to six switches (see ez switch lib Example 4)
 - two x button switches wired with a 10k ohm pull down resistor
 - one x button switches wired without a 10k ohm pull down resistor
 - two x toggle switches wired without a 10k ohm pull down resistor
 - one x toggle switches wired with a 10k ohm pull down resistor

but, there is no reason to believe that more could not be configured within the limits of the chosen microcontroller.

- 3. For every switch configured, 12 bytes of free memory will be allocated at run time when the <ez_switch_lib> class is initiated. This memory requirement is in addition to the size of the compiled sketch.
- 4. The period of time defined for switch noise debounce is global and applicable to all switches, irrespective of type. It is preset 'out of the box' (OOTB) to 10 milliseconds but it may be programmatically adjusted by the end user code, as required (see function set_debounce, below).
- 5. The library supports two simple and commonly seen switch wiring schemes (see Common Switch Wiring Schemes), these being without the use of any hardware components other than, optionally, 10k ohm resistors. Even these can be dispensed with! Having said that, the library should also support (but not tested) switches connected with hardware debounce circuits. If this is the case then set the software debounce period to 0 milliseconds (see function set debounce, below).
- 6. For switches to be responsive in something like real-time, they need to be tested frequently and, for button switches particularly, processed when a switch cycle is detected. However, toggle switches may have their current status examined at any point and any time. A software design based on a switch polling loop should be an ideal harness to ensure continuous switch testing and processing.

Switch Types Supported

There are so, so many switches available, many for specific purposes but most of a general nature and suitable for the majority of needs.

The switch library was developed to support two types of common and general use switches – button, or momentary switches, and toggle switches. Of course this latter type of switch, toggle, itself comes in all kinds of designs, for example, simple single lever, pop-on pop-off, rotary, slide, tilt, etc.



Examples of common types of switch


The principal distinction between button (momentary) and toggle type switches is that button switches have a switch cycle of OFF-ON-OFF which signifies switch activation, whereas toggle switches go through either OFF-ON or ON-OFF representing two distinct and

RDB Stafford,UK - 6 - v1.03, March 2021

separate switch transitions. The status of toggle switches therefore persists after being physically switched – they stay ON or OFF. <ez_switch_lib> automatically handles these physical characteristics.

Common Switch Wiring Schemes

If you now appreciate the differences between switch types, it is necessary to understand how they should be connected to the microcontroller and the differences between the two commonly seen switch wiring schemes <u>without</u> hardware debounce. These are shown below, as 'circuit C1' and 'circuit C2':



<u>Either</u> circuit can be used for <u>either</u> type of switch, but the key to configuring correctly lies in the way they are software configured via the <u>pinMode</u> function, as follows.

- The pinMode setting for initialising circuit_C1 is pinMode (<pin>, INPUT). This has the effect of setting the digital pin <pin> to 0v, representing 'off'. The 10k ohm pull down resistor is essential and ensures that the pin stays at 0v until switched, otherwise the input pin will be susceptible to spurious firing from extraneous fields. When the switch is actuated the input rises to +5v which will be detected as the switch transitioning to 'on'.
- For circuit_C2 the pinMode setting is pinMode (<pin>, INPUT_PULLUP). This brings into play an internal microcontroller pull up resistor resulting in the digital pin floating at 5v, representing 'off'. No external resistor is required and when the switch is actuated the pin will be brought to 0v which will be detected as the switch transitioning to 'on'.

Note the reversed conditions for 'off' and 'on' between the two circuit schemes. The <ez switch lib> will account for this automatically.

RDB Stafford,UK -7 - v1.03, March 2021

Using the <ez switch lib> Library

To Know...

The switch reading functions of the library <ez_switch_lib> are functions that read the Arduino's digital pins as inputs. At the heart of these functions lay the use of the general digitalRead function of the <Arduino.h> library. So, if that is the case, why do we need other functions to read digital pins?

Well, there are several reasons:

- not all switches are the same, there is a difference between how button and toggle switches behave
- when switches transition to on/off, off/on (toggle switches) or off/on/off (button switches) they do so in a small finite time during which they will often generate 'noise' which can be read as a spurious signal and confuse the calling code logic
- switches may be wired in different ways which can reverse the logical understanding what the meaning of LOW and HIGH is.

In short, and as previously stated, switches can be troublesome for the uninitiated! The <ez_switch_lib> library switch read functions deal with all of the above issues, so we can concentrate purely on what we want switches to do for us.

Both the digitalRead and the $\langle ez_switch_lib \rangle$ library switch read functions test a digital pin at a point in time and in a non-blocking way (they have a 'quick look' and immediately move on – returning back to the calling code). The downside of non-blocking functionality means that digital pins with switches connected must be regularly tested if switch actuations are not to be missed. In fact, they should all be tested at every opportunity. If you look at the Example Sketches 1.1-4, they are all designed to constantly poll every switch at every main loop cycle. It is in this way that we are able to ensure the code can keep up with what is happening the real world, without missing a switch actuation.

There are three <ez_switch_lib> library functions we can use to read a switch, once it is declared to the library class. These are:

- read button switch
- read toggle switch
- read switch

All have the same single parameter (switch id) and return values (see Specifications - Switch Control Functions).

The first two functions are obvious in their purpose. The third function, read_switch, is 'agnostic' and may be used without regard to what type of switch is connected. It should be the principal read function in use in your sketch.

All of these functions return a value of switched or !switched (not switched), a little like the digitalRead return values of HIGH and LOW.

However, the <ez_switch_lib> library functions provide additional information that is useful to the programmer:

• For <u>button switches</u>, the programmer can access a library variable that will indicate if a button switch is in transition (off-on-off cycle active and pending completion). The library variable is <class name>.switches[switch_id].switch_pending, where <class name> is the name you have given to the library's <u>Switches</u> class when you declared/initiated it. The permissible values of this variable are either true or false. For example:

```
my_switches.read_switch(button_id); // test switch
if (my_switches.switches[button_id].switch_pending == true) {
    // switch is in transition, so take it as on until not pending...
}
```

 For toggle switches, we also have access to the same switch pending variable as above, but also an additional one. Recall that toggle switches are either one state (on/off) or the other until they are actuated (flipped). We therefore need to be able to test what the current/now setting of a toggle switch may be, that is, its current status. We may similarly do this as follows:

```
my_switches.read_switch(toggle_id); // test switch
if (my_switches.switches[toggle_id].switch_status == on) {
    // switch is currently on...
}
```

The permissible values of this variable are on or !on (off).

See Example 5 which shows a sketch employing the above to implement a routine to increment hours and minutes of an external clock/timer, working both in single-shot and continuous advance modes.

Finally, the other two <code><ez_switch_lib></code> library functions - <code>read_button_switch</code> and <code>read_toggle_switch</code>, will operate in the same way as <code>read_switch</code> but specifically for each type of switch only. In addition, if you use these two read functions and your switches have linked outputs then these outputs will <code>not</code> be processed. Only the <code>read_switch</code> function will process any linked switch output (see Specifications - Switch Control Functions).

Hopefully, you are now better informed to make good use of the <ez_switch_lib> library and its data structures and functions, all of which are documented below – see Specifications and Example Sketches.

Location of the <ez switch lib> Library

The switch library files should be installed within a directory called 'ez_switch_lib' under the Arduino libraries directory - ...\Arduino\libraries\.

The <ez switch lib> directory will comprise four files:

```
    ez_switch_lib.h
    ez_switch_lib.cpp
    C++ functions
```

- 3. keywords.txt ... keyword file to highlight <ez switch lib> keywords
- 4. ez_switch_lib_user_guide.pdf ... this document, or file elsewhere if required
- 5. ez_switch_lib crib sheet.pdf ... an easy reference for key library data/functions, or file elsewhere if required

Steps to Successful Use

Before 'flighting to task', it is recommended to think carefully about what it is you wish to achieve, how switches are incorporated into your project and how <ez_switch_lib> can be utilised.

The principal considerations are:

- 1. Decide how many switches and of what type these will be.
- 2. For each switch decide
 - a. which digital pin will be used?
 - b. how will the switch be wired, circuit C1 or circuit C2?
 - c. do we wish to create a link from the switch to a digital output pin and, if so, what state should the output pin be initialised to at creation time?
- 3. What will happen when each switch is activated? This step is beyond this UG and is the purpose of your project.

If you are implementing many switches then it may be helpful to make a note of their configurations as once you start wiring and coding things can get a bit muddled up! The following template may be helpful to fill out at the start of your planning and for you to refer to into the development stage (it is also a useful documentation aid post implementation):

F	Project Name:							Date:
	Switch Configs			Linked Outputs				
Pin	Switch	n Type	Circuit	Туре	Pin	Initial Value		Notes
FIII	Button	Toggle	C1	C2	FIII	LOW	HIGH	

(add more rows as needed)

For example, Example 4, below, configures the following switches, pins, circuits and links:

F	Project Name: LEDs & Relays Date: 4 March 2021									
	Switch Configs					nked Out	puts			
Pin	Switch	n Type	Circuit	Туре	Pin	Initial	Value	Notes		
FIII	Button	Toggle	C1	C2		LOW	HIGH			
2		Χ	Χ		8		Χ	Relay 1 -	no switching	coding
3		Χ		Χ	9	Χ		Led 1 -	no switching	coding
4		Χ	Χ					Led 3 -	needs switch	ing coding
5	Х			Х				Produces switch status report to serial		eport to serial
								monitor		
6	Χ		Χ		10		Χ	Relay 2 -	no switching	coding
7	Х			Х	11	Х		Led 2 -	no switching	coding

Having got to grips with what switches, pins and circuit schemes your project will be designed around it is necessary to understand how <ez_switch_lib> can be used. As with all libraries there are a number of points to consider:

- 1. We need to ensure our sketch references the library
- 2. We need to create an instance of the library class, and
- 3. We need to understand how to correctly use the library's capabilities (e.g. functions and data).

There are a number of steps to be followed –

Step 1

To start, we need to declare the <ez_switch_lib> library. At the top level of your sketch include the following statements:

```
#include <Arduino.h>
#include <ez switch lib.h>
```

Step 2

Then, prior to <code>setup()</code>, declare how many switches your sketch will be configured for (e.g. <code>#define num_switches 6</code>, or <code>byte num_switches = 6</code>; etc) together with your switch configuration data (as per the above template?). How you wish to declare this data is very much up to your personal preference. The example sketches, below, show several approaches that you may find instructive.

Step 3

Again, <u>prior</u> to your sketch setup() function and after your switch data declarations, add the following class initiation statement:

```
Switches my switches (num switches);
```

Where 'my_switches' is the name you wish to use for the class you have initiated – this can be anything, but 'my switches' is a pretty good name.

Step 4

Okay, we're off and running? Not quite, before we can plough on and start reading switches we need to declare them to the library along with their attributes. We do this by using the function add.switch. This function will add a specified switch to the library's table of active switches such that when it is read (tested) by the read function(s) it will know how it is to be handled. Therefore for <u>each</u> switch you wish to configure you will need to add it to the library's active switch table; for this we use the add.switch function:

```
byte switch_id;
switch_id = my_switches.add.switch(button_switch, 4, circuit_C2);
```

Things to notice:

 the add.switch function call is preceded with the name we have given to the Switches class, in this example 'my_switches'. This is required to access any resource within the class

- 'button_switch' and 'circuit_C2' are reserved keywords and are highlighted in red.
 They each define the switch and circuit type, respectively. There are a number of
 reserved words you may use throughout your sketch, see Specifications Reserved
 Macro Definitions and Switch Control Functions add.switch to understand the possible
 parameters and return values/conditions
- the function provides a return value. If the addition is successful, this value is the
 reference you should use whenever you use any of the library resources where switch
 reference is a required parameter. How you retain this is very much up to your design,
 but see the example sketches below which should prove helpful. See Specifications Switch Control Functions add.switch to understand the possible return
 values/conditions.

The best place to add/create your switches is in the setup() function, but it can be done anywhere so long as it is only done once and is in scope of the library class statement.

Step 5

Now that is done we can start to read the switches.

The simplest way to read a switch is to use the function <code>read_switch</code>. This function is agnostic to switch type and has a single parameter - the id of the switch we wish to read. It will return either 'switched' or '!switched' (again reserved library macros), the meaning being obvious. For example:

```
if(my_switches.read_switch(switch_id) == switched)
{
...do something;
}
```

There are other resources available from the library and these are described below.

Example - to recap the steps (five) in order of application/use are:

Step	Example
1	#include <arduino.h></arduino.h>
	<pre>#include <ez_switch_lib.h></ez_switch_lib.h></pre>
	// plus any other libraries
2	#define num_switches 1 // number of switches to be added, 1 in this example
	// define your switch data
	<pre>byte switch_id;</pre>
	<pre>byte button_pin = 4;</pre>
3	// create switch class instance
	<pre>Switches my_switches(num_switches);</pre>
4	<pre>void setup() {</pre>
	<pre> // declare your switches to the library, for example: switch_id = my_switches.add.switch(button_switch, button_pin,circuit_C2); // validate return value }</pre>
5	<pre>void loop(){</pre>
	do{
	<pre>if(my switches.read switch(switch id) == switched)</pre>

```
Step Example
{
    ...do something;
} while (true);
}
```

Additionally to note:

If you need to reference any of the library's class resources then you must prefix them with the name you have given to the class when you created/initiated it. For example:

```
my_switches.switches[switch_id].switch_type,
my_switches.switches[2].switch_status,
my_switches.switches[toggle_id3].switch_pending,
my_switches.add_switch(button_switch,12,circuit_C2),
my_switches.num_free_switch_slots(),
my_switches.set_debounce(25),
my_switches.read_switch(my_switch_data[sw]),
switch_id = my_switches.last_switched_id
etc.
```

Specifications

Specifications – Switch Control Structure (SCS)

At the heart of the <ez_switch_lib> library lies a struct(ure) 'table' - the switch control structure (SCS), that is used to hold the data attributes for all declared/defined switches.

At initiation of the class, the SCS is created from free memory using a malloc call of sufficient size to match the number of switches the class is being defined for. Thereafter, it may be populated with switches by use of the add_switch function (see below) up to the maximum number of switches declared for the class.

The SCS has the following construction and layout:

Members of the SCS may be directly accessed from the end user sketch, as required, see above. Of particular interest will be:

For button and toggle switches -

```
my switches.switches[switch id].switch pending
```

and for toggle switches only -

my switches.switches[2].switch status

Specifications - Reserved Macro Definitions and Other Declarations

The table below documents the library's reserved macro definitions. <u>These are available for use by a sketch</u> simply by referencing their name (column 1 and no prefix required), see example sketches. When used they will be coloured in red to show that they are reserved words.

Macro definitions	Values	Significance / Comments
#define button_switch	1	differentiates switch type, this being of type
		'button'
<pre>#define toggle_switch</pre>	2	differentiates switch type, this being of type
		'toggle'
#define circuit_C1'	INPUT	switch circuit configured with an external
		pull down 10k ohm resistor
<pre>#define circuit_C2</pre>	INPUT_PULLUP	switch circuit configured without an external
	4	pull down resistor
#define switched	true	A value returned by read_switch,
		read_button_switch &
		read_toggle_switch functions.
		Cinnifica quitab bas base present and
		Signifies switch has been pressed and
		switch cycle complete, otherwise !switched
#define on	true	used for toggle switch status. Off is !on
#define not used	true	'not used' indicator – marks if a field in the
"	3243	switch control structure is used or not
#define bad params	-2	A value returned by add switch function -
		invalid parameters
#define add_failure	-1	A value returned by add switch function -
		could not insert a given switch, i.e. no slots
		left
#define link_success	0	A value returned by
		link_switch_to_output function -
		output pin successfully linked to a switch
<pre>#define link_failure</pre>	-1	A value returned by
		<pre>link_switch_to_output function -</pre>
		output pin could not be linked to a switch
<pre>#define none_switched</pre>	255	last_switched_id variable initialised to
		this value until first switch is actuated

Other Declarations

The library supports a useful variable, accessible to the end user developer, that records the id of the switch that has last been actuated, i.e. switched. This variable is declared and initialised as follows:

byte last switched id = none switched;

And may be referenced as follows (example):

```
my switches.last switched id
```

See the example of its use in the Corollary section below.

Specifications - Switch Control Functions

Type	int Name add_switch							
Parameters	<pre>byte sw_type, byte sw_pin, byte circ_type parameter choices are: sw_type - is either 'button_switch' or 'toggle_switch', sw_pin - is the digital pin assigned to the switch, circ type - is either 'circuit C1'' or 'circuit C2'.</pre>							
Purpose / functionality	This function will add (create) the specified switch (parameters) to the switch control structure, if possible.							
	There are three possible outcomes from an add_switch call:							
	 Successful addition of switch. In this case the return value is >= 0 and represents the physical slot (location 'switch_id/token') of the switch in the switch control structure. This should be retained by the calling code/design. No further slots available in the switch control structure, all are used. The supplied parameters are 'bad'. 							
	The results of an add switch call are as below.							
Return values	Return values are:							
	>= 0 success, switch added to switch control struct(ure) - the switch control structure entry number is returned (switch_id/token) for the switch added, -1 add_failure - no slots available in the switch control structure, -2 bad_params - given parameter(s) for switch are not valid.							
Example								
	Example							

```
void create_my_switches() {
 for (int \overline{sw} = 0; sw < num switches; sw++) {
   int switch_id =
     my_switch_data[sw][2]);// circuit type
   if (switch id < 0)
   { // There is a data compatibility mismatch (-2),
     // or no room left to add switch (-1).
     Serial.print("Failure to add a switch:\nswitch entry:");
     Serial.print(switch_id);
     Serial.print(", data line = ");
     Serial.print(my_switch_data[sw][0]);
Serial.print(", ");
     Serial.print(my_switch_data[sw][1]);
Serial.print(", ");
     Serial.println(my_switch_data[sw][2]);
     Serial.println("!! PROGRAMME TERMINATED !!");
     Serial.flush();
     exit(1);
    } else {
     // 'switch_id' is the switch control slot entry for this switch (sw),
      // so we can use this, if required, to know where our switches are
```

Туре	int Name link_switch_to_output						
Parameters	<pre>byte switch_id, byte output_pin, bool HorL</pre>						
Purpose / functionality	This function will link the given digital pin (output_pin) to an already created/defined switch (switch_id) and initialise it according to the specified parameter, as follows:						
	<pre>pinMode (output_pin, HorL), where HorL is either LOW or HIGH</pre>						
	Once linked, the output pin will be flipped between LOW/HIGH or HIGH/LOW each time the associated switch is read (tested) by the <pre>read_switch</pre> function and found to have been actuated (switched).						
	 Note that: This feature allows simple digital output pin switching without any requirement for end user coding. The output can be initialised at either LOW or HIGH level. Automatic output pin flipping only occurs via use of the read_switch function. If switches have a linked output and they need to be read without automatic output pin flipping then use read_button_switch / read_toggle_switch, instead. These functions will not flip associated switch output levels. Redefining a linked switch output - existing defined switch linked outputs can be redefined, as required, by further calls to the switch link function. Removing a linked switch output - if a switch has a linked output defined and it is necessary to remove it then this can be done by a call to link_switch_to_output with an output pin value of 0. The output level will be set according to the Horl parameter. For example: 						
	<pre>link_result = my_switches.link_switch_to_output(</pre>						
	However, if the output level is to remain unaltered then set the Horl						
	<pre>parameter to 'my_switches'.switches[switch_id].</pre>						
	<pre>switch_out_pin_status, where 'my_switches' is the name of your class for the Switches class.</pre>						
	For example:						
	<pre>link_result = my_switches.link_switch_to_output(</pre>						
	<pre>my_switches.switches[switch_id].</pre>						
Return values	0, link_success - linkage was successful -1, link_failure - linkage failed, switch_id not in range of defined						

```
link switch to output
Type
                 int
                         Name
                                       switches
                                         Example
Example 1:
  // Link/associate this switch to the in-built led (normally pin 13)
  // with the switch we have just installed/created so that every
  // time the switch is actuated the built in LED will be automatically
  // flipped. Start with LED at LOW setting.
 int link result = my switches.link switch to output(
                      switch id,
                      LED BUILTIN,
                      LOW);
 if (link result == link failure ) {
    // linking failed, invalid switch id
    Serial.begin(9600);
    Serial.println(F("Failure to link an output to a switch"));
    Serial.println(F("!!PROGRAM TERMINATED!!"));
    Serial.flush();
    exit(2);
  }
Example 2:
  // Link/associate this switch to the relay output pin
  // with the switch we have just installed/created so that every
  // time the switch is actuated the relay will be automatically
  // flipped. Start with relay at HIGH setting.
  int link result = my switches.link switch to output(
                      switch id,
                      relay_{1},
                      HIGH);
 if (link result == link failure ) {
    // linking failed, invalid switch id
    Serial.begin(9600);
    Serial.println(F("Failure to link an output to a switch"));
    Serial.println(F("!!PROGRAM TERMINATED!!"));
    Serial.flush();
    exit(2);
```

Туре	int	Name	num_free_switch_slots				
Parameters	none						
Purpose / functionality	Returns	Returns the number of free slots available in the switch control structure.					
Return values	0 to the maximum number of switches defined						
Example							
<pre>Serial.print("\nNumber of free switch slots in the SCS = "); Serial.println(my_switches.num_free_switch_slots());</pre>							

Туре	bool	Name	read_switch							
Parameters	byte switch id									
Purpose /	Functio	n will rea	nd the given switch returning a result as below.							
functionality										
	Note th	at:								
	1. The	switch	id parameter is the switch entry number in the switch control							
	structure of the switch to be read. This is the returned value from the									
	add switch function call.									
	2. If a	n invalid	switch_id is given the read function exits with a return value							

Туре	bool Name read_switch										
	3. If a	 of !switched. 3. If a switch has a linked output pin associated with it then this function will flip the current output pin level, i.e. from Low to HIGH, or from HIGH to Low. See add_switch and link_switch_to_pin for further information. 									
Return values	switch	ned or !s	switched								
			Example								
<pre>// need to hand do { if (my_switc) led_level : digitalWri }</pre>	<pre>// the switch does not have and output pin linked to it, so we // need to handle the flipping of the LED do { if (my_switches.read_switch(switch_id) == switched) { led_level = HIGH - led_level; // flip between HIGH and LOW each cycle digitalWrite(LED_BUILTIN, led_level);</pre>										
// connected to // We therefore // LED will be do {	<pre>Example 2: // the switch has been defined with a linked/associated output pin // connected to a LED. // We therefore have nothing to do but keep reading the switch and the // LED will be automatically flipped for us. do { my_switches.read_switch(switch_id); } } </pre>										

Type	bool Name read button switch									
Parameters	byte switch_id									
Purpose / functionality	This is used by the <pre>read_switch</pre> function and deals specifically with reading momentary button style switches. The function can be used by end user code, but note: 1. Remember that the <pre>switch_id</pre> parameter is the switch entry number in the switch control structure of the switch to be read. 2. If the switch has a linked/associated output pin then it will <pre>not</pre> be processed.									
Return values	switched or !switched									
	Example									
<pre>if (my_switches.read_button_switch(switch_id) == switched) { // button switch pressed }</pre>										

RDB Stafford,UK -18 - v1.03, March 2021

Туре	bool Name read_toggle_switch										
Parameters	byte switch_id										
Purpose / functionality	This is used by the <pre>read_switch</pre> function and deals specifically with reading toggle style switches. The function can be used by end user code, but note: 1. Remember that the <pre>switch_id</pre> parameter is the switch entry number in the switch control structure of the switch to be read. 2. If the switch has a linked/associated output pin then it will <pre>not</pre> be processed.										
Return values	switched or !switched										
	Example										
<pre>if (my_switches.read_toggle_switch(switch_id) == switched) { // toggle switch switched }</pre>											

Туре	void Name print_switch										
Parameters	byte switch_id										
Purpose / functionality	switch	The function prints the switch parameters of the switch defined at slot switch_id in the switch control structure to the serial monitor. It can be helpful in the debugging phase and removed thereafter.									
Return values	none										
			Example								
my_switches.prin	<pre>my_switches.print_switch(3);</pre>										
<pre>Example output a toggle switch, configured as circuit_C2 and occupying slot 3 (switch_id = 3) in the switch control structure:</pre>											
<pre>slot: 3 sw_type= 2 sw_pin= 5 circ_type= 2 pending= 0 db_start= 0 on_value= 0 sw_status= 0</pre>											

Туре	void Name print_switches										
Parameters	none										
Purpose / functionality	The function prints the switch parameters of ALL switches held in the switch control structure to the serial monitor. It can be helpful in the debugging phase and removed thereafter.										
Return values	none										
	Example										
<pre>my_switches.print_switches();</pre>											
Example output for 6 defined switches - 3 \times button & 3 \times toggle, configured as either circuit_C1 or circuit_C2:											
slot: 0 sw_type= 1 sw_pin= 2 circ_type= 0 pending= 0 db_start= 0 on_value= 1 sw_status= 1 op_pin= 8 op_status= 0											
slot: 1 sw_type=	<pre>sw_pin= 3 circ_type= 2 pending= 0 db_start= 0 on_value= 0 sw_status= 1</pre>										

```
Type void Name print_switches

op_pin= 9 op_status= 1

slot: 2 sw_type= 1 sw_pin= 4 circ_type= 0 pending= 0 db_start= 0 on_value= 1 sw_status= 1

op_pin= 0 op_status= 0

slot: 3 sw_type= 2 sw_pin= 5 circ_type= 2 pending= 0 db_start= 0 on_value= 0 sw_status= 0

op_pin= 0 op_status= 0

slot: 4 sw type= 2 sw pin= 6 circ type= 0 pending= 0 db start= 0 on value= 1 sw status= 0

op_pin= 0 op_status= 0

slot: 5 sw_type= 2 sw_pin= 7 circ_type= 2 pending= 0 db_start= 0 on_value= 0 sw_status= 0

op_pin= 0 op_status= 0
```

Туре	void	void Name set_debounce									
Parameters	int pe	int period									
Purpose / functionality	Note th	eading for at: debounce debounce	e value is set to 10 milliseconds, by default e setting is global and applies to ALL defined switches er value must be >= 0. Negative values are ignored.								
Return values	none										
Example											
<pre>my_switches.set_debounce(20); // set debounce for 20 msecs</pre>											

RDB Stafford,UK - 20 - v1.03, March 2021

Example Sketches

What follows are a number of examples in the use of the $<ez_switch_lib>$ library. These are provided to aid understanding in how the $<ez_switch_lib>$ can be applied to your projects.

Each example sketch may be copied and pasted directly into the Arduino IDE from the github links provided for each example sketch, compiled and uploaded without any further coding – just ensure that you have downloaded the <ez switch lib> library files first.

Any additional components beyond an Arduino microcontroller, connecting wires and a breadboard are indicated for each sketch.

The example sketches are:

- 1. Example 1.1 turning on and off the in-built LED of the Arduino microcontroller (normally on pin 13) using a button switch using direct coding.
- 2. Example 1.2 as for example 1.1 but using the function link_switch_to_output to flip the in-built LED using indirect coding.
- 3. Example 2,1 turning on and off the in-built LED of the Arduino microcontroller (normally on pin 13) using a <u>toggle</u> switch using direct coding.
- 4. Example 2.2 as for example 2.1 but using the function link_switch_to_output to flip the in-built LED using indirect coding.
- 5. Example 3.1 four switches, two button and two toggle, wired in different schemes, with each switch turning on and off an associated LED using direct coding.
- 6. Example 3.2 as for example 3.1 but using the function link_switch_to_output to flip each of the associated LEDs using indirect coding.
- 7. Example 4 six switches, three button and three toggle, wired in different schemes, with each switch being processed by its own switch-case statement. The sketch incorporates a mix of direct coding and use of the function link_switch_to_output to control the effects of the switches, using the serial monitor, LEDs and relays.

All of the sketches can be accessed from github - follow the specific github link with each example. Alternatively, the main <ez_switch_lib> github page with the <ez_switch_lib> files (.h, .cpp, .txt and .pdf files) can be found at this link.

RDB Stafford,UK - 21 - v1.03, March 2021

Example 1.1 - Turning LED On/Off With a Button Switch, Directly Coded

This example sketch uses a button switch and will turn the Arduino in-built led on and off with each press. The switch is wired for circuit C1.

Note that a led state change will only occur when the button switch is released, that is after the completion of the switching cycle.

The switch mappings and outputs are:

	Project Name: Example 1.1 – button switch, no linking									4 March 2021
Switch Configs Linked Outputs										
Pin	Switch	п Туре	pe Circuit Type			Initial Value		Notes		
FIII	Button	Toggle	C1	C2	Pin	LOW	HIGH			
2			>					No linked output, but	ton will fl	ip in-built LED
	^		^					by direct coding		

Components required	Circuit schemes
1 x button switch	_circuit_C1
	Ov (ground) 10k ohm resister Digital read pin switch
	Arduino Microcontroller Figure 1 – Circuit C1

The sketch can be also accessed from github, here.

```
Ron D Bentley, Stafford, UK
  Mar 2021
  Example of use of the ez_switch_lib library
  Example 1.1
  Reading single button switch to turn built in led on/off.
  When the button switch is actuated, the in-built led
  (LED_BUILTIN), will be flipped ON/OFF etc by using suitable
  coding in the sketch's main loop.
  This example and code is in the public domain and
  may be used without restriction and without warranty.
#include <Arduino.h>
#include <ez switch lib.h> // ez switch lib .h & .cpp files are stored under
...\Arduino\libraries\ez switch lib\
int switch id;
bool led level = LOW; // start with led off
#define num switches 1 // only a single switch in this sketch example
// Declare/define the switch instance of given size
Switches my_switches(num_switches);
void setup() {
 // Attach a button switch to digital pin 2, with
 // an external pull down resistor, circuit C1,
 // and store the switch's id for later use.
 switch id = my switches.add switch(
              button switch,
```

```
circuit C1);
  // validate the return
 if (switch_id < 0) {</pre>
   // Error returned - there is a data compatibility mismatch (-2, bad params),
    // or no room left to add switch (-1, add failure).
   Serial.begin(9600);
    Serial.println(F("Failure to add a switch"));
    if (switch id == add failure) {
     Serial.println(F("add_switch - no room to create given switch"));
    } else {
      // Can only be that data for switch is invalid
      Serial.println(F("add switch - one or more parameters is/are invalid"));
   Serial.println(F("!!PROGRAM TERMINATED!!"));
   Serial.flush();
    exit(1);
 // Initialise built in led and turn to off
 pinMode(LED_BUILTIN, OUTPUT);
 digitalWrite(LED_BUILTIN, LOW);
void loop() {
 // Keep reading the switch we have created and toggle the built in
  // led on/off for each press.
 do {
   if (my switches.read switch(switch id) == switched) {
     // Flip between HIGH and LOW each cycle
     led level = HIGH - led level;
     digitalWrite(LED BUILTIN, led level);
  } while (true);
```

Example 1.2 - Turning LED On/Off With a Button Switch, Indirectly Coded

This example sketch uses a button switch to turn the Arduino in-built led on and off with each press, indirectly, by using the <code>link_switch_to_output</code> function. Compare this sketch with example 1.1 sketch and note the differences – the button switch is linked to an output pin and no code exists in the sketch to flip the output pin (in-built LED). The switch is wired for <code>circuit C1</code>.

Note that a led state change will only occur when the button switch is released, that is after the completion of the switching cycle.

The switch mappings and outputs are:

Project Name: Example 1.2 – button switch, with linking										4 March 2021	
Switch Configs Linked Outputs											
Pin Switch Type Circuit Type				Туре	Pin	Initial Value		Notes			
FIII	Button	Toggle	C1	C2	FIII	LOW	HIGH				
2	Х		Х		13	Х		LED_BUILTIN – to flip the LED	 no direct coding needed 		

Components required	Circuit schemes
1 x button switch	_circuit_C1
	Ov (ground) 10k ohm resister Digital read pin switch
	Arduino Microcontroller Figure 1 – Circuit C1

The sketch can be accessed from github here.

```
Ron D Bentley, Stafford, UK
  Mar 2021
  Example of use of the ez switch lib library
  Reading single button switch to turn built in led on/off, but
  in this example we shall link the switch to an output pin
  (LED_BUILTIN) using a ez_switch_lib function, so that when
  actuated, the output pin will be automatically flipped
  HIGH-LOW etc each time the button switch is pressed WITHOUT
  any further coding.
  This example and code is in the public domain and
  may be used without restriction and without warranty.
#include <Arduino.h>
#include <ez switch lib.h> // ez switch lib .h & .cpp files are stored under
...\Arduino\libraries\ez switch lib\
int switch id;
#define num_switches 1 // only a single switch in this sketch example
// Declare/define the switch instance of given size
Switches my switches (num switches);
void setup() {
```

```
// Attach a button switch to digital pin 2, with
  // an external pull down resistor, circuit C1,
  // and store the switch's id for later use.
 switch_id = my_switches.add_switch(
                button switch,
                2,
                circuit C1);
  // validate the return
 if (switch id < 0) {
    // Error returned - there is a data compatibilty mismatch (-2, bad_params),
    // or no room left to add switch (-1, add failure).
   Serial.begin(9600);
    Serial.println(F("Failure to add a switch"));
    if (switch id == add failure) {
     Serial.println(F("add switch - no room to create given switch"));
    } else {
      // can only be that data for switch is invalid
     Serial.println(F("add_switch - one or more parameters is/are invalid"));
   Serial.println(F("!PROGRAM TERMINATED!!"));
    Serial.flush();
   exit(1);
  // Link/associate this switch to the in-built led (normally pin 13)
  // with the switch we have just installed/created so that every
  // time the switch is actuated the built in LED will be automatically
  // flipped. Start with LED at low setting.
 int link result = my switches.link switch to output(
                      switch id,
                      LED BUILTIN,
                      LOW);
 if (link result == link failure ) {
    // linking failed, invalid switch id
    Serial.begin(9600);
    Serial.println(F("Failure to link an output to a switch"));
    Serial.println(F("!!PROGRAM TERMINATED!!"));
    Serial.flush();
    exit(2);
  }
}
void loop() {
    // just keep reading, LED_BUILTIN will automatically be flipped for us
   // so we dont need to do anything else
   my switches.read switch (switch id);
 while (true);
```

Example 2.1 - Turning LED On/Off With a Toggle Switch, Directly Coded

This example sketch uses a toggle switch and will turn the Arduino in-built led on and off with each actuation. The sketch is essentially the same as example 1.1, the difference being that a toggle switch type is declared instead of a button type. The switch is wired for circuit C1.

Note that a led state change occurs at <u>each</u> position of the toggle switch.

The switch mappings and outputs are:

	Project Name: Example 2.1 – toggle switch, no linking									4 March 2021
Switch Configs Linked Outputs							puts			
Pin	Switch Type Circuit Type			Туре	Pin Initial Value		Value	Notes		
FIII	Button	Toggle	C1	C2	FIII	LOW	HIGH			
2		>						No linked output, t	oggle will	flip in-built LED
2		X		^				by direct coding		

Components required	Circuit schemes
1 x toggle switch	circuit_C1
	(ground) 10k ohm resister
	Digital read pin
	+5v pin II
	Arduino Microcontroller Figure 1 – Circuit C1

The sketch can be accessed from github here.

```
Ron D Bentley, Stafford, UK
  Mar 2021
  Example of use of the ez switch lib library
  Example 2.1
  Reading single toggle switch to turn built in led on/off.
  When the toggle switch is activated, the in-built led
  (LED BUILTIN), will be flipped ON/OFF etc by using suitable
  coding in the sketch's main loop.
  This example and code is in the public domain and
  may be used without restriction and without warranty.
*/
#include <Arduino.h>
#include <ez_switch_lib.h> // ez switch lib .h & .cpp files are stored under
...\Arduino\libraries\ez switch lib\
int switch id;
bool led level = LOW;
#define num switches 1 // only a single switch in this sketch example
// Declare/define the switch instance of given size
Switches my switches (num switches);
void setup() {
 // Attach a toggle switch to digital pin 2, with
 // an external pull down resistor, circuit_C1,
 // and store the switch's id for later use.
```

```
switch id = my switches.add switch(
                toggle_switch,
                circuit C1);
  // Validate the return
 if (switch id < 0) {
    // Error returned - there is a data compatibilty mismatch (-2, bad params),
    // or no room left to add switch (-1, add failure).
    Serial.begin(9600);
    Serial.println(F("Failure to add a switch"));
    if (switch id == add failure) {
     Serial.println(F("add switch - no room to create given switch"));
     // Can only be that data for switch is invalid
     Serial.println(F("add switch - one or more parameters is/are invalid"));
   Serial.println(F("!!PROGRAM TERMINATED!!"));
   Serial.flush();
   exit(1);
  // Initialise built in led and turn to off
 pinMode(LED_BUILTIN, OUTPUT);
 digitalWrite(LED BUILTIN, LOW);
void loop() {
 // keep reading the switch we have created and toggle the built in
 // led on/off for each press.
    if (my switches.read switch(switch id) == switched) {
     // flip between HIGH and LOW each cycle
      led level = HIGH - led level;
     digitalWrite(LED_BUILTIN, led_level);
  } while (true);
```

Example 2.2 - Turning LED On/Off With a Toggle Switch, Indirectly Coded

This example sketch uses a toggle switch to turn the Arduino in-built led on and off, indirectly, by using the link_switch_to_output function. It is essentially the same sketch as in example 2.1, above. Compare this sketch with example 2.1 sketch and note the differences – the toggle switch is linked to an output pin and no code exists in the sketch to flip the output pin (in-built LED).

Note that a led state change occurs at <u>each</u> position of the toggle switch.

The switch mappings and outputs are:

F	Project Na	ame: Ex	ample 2	2.2 – to	Date: 4 March 202	1			
	Switch Configs			Linked Outputs		puts			
Pin	Switch	n Type	Circuit	Туре	Pin	Din Initial Value		Notes	
PIII	Button	Toggle	C1	C2	FIII	LOW	HIGH		
2		Х	Х		13		Х	LED_BUILTIN – no direct coding needed to flip the LED	d

Components required	Circuit schemes
1 x toggle switch	_circuit_C1
	(ground) 10k ohm
	Digital read pin
	+5v pin
	Arduino Microcontroller Figure 1 – Circuit C1

The sketch can be accessed from github here.

```
Ron D Bentley, Stafford, UK
  Mar 2021
  Example of use of the ez switch lib library
  Example 2.2
  Reading single toggle switch to turn built in led on/off.
  Toggle switch is associated with an output pin (LED BUILTIN)
  using a ez switch lib function, so that when activated, the
  output pin will be automatically flipped, HIGH-LOW etc each
  time the toggle switch is actuated WITHOUT any further coding.
  This example and code is in the public domain and
  may be used without restriction and without warranty.
#include <Arduino.h>
#include <ez switch lib.h> // ez switch lib .h & .cpp files are stored under
...\Arduino\libraries\ez switch lib\
int switch id;
#define num switches 1 // only a single switch in this sketch example
// Declare/define the switch instance of given size
Switches my switches (num switches);
void setup() {
 // Attach a toggle switch to digital pin 2, with
 // an external pull down resistor, circuit C1,
```

```
// and store the switch's id for later use.
  switch id = my switches.add switch(
                toggle switch,
                circuit C1);
  // Validate the return
 if (switch id < 0) {
    // Error returned - there is a data compatibilty mismatch (-2, bad params),
    // or no room left to add switch (-1, add failure).
    Serial.begin(9600);
    Serial.println(F("Failure to add a switch"));
    if (switch id == add failure) {
     Serial.println(F("add_switch - no room to create given switch"));
      // Can only be that data for switch is invalid
      Serial.println(F("add switch - one or more parameters is/are invalid"));
   Serial.println(F("!!PROGRAM TERMINATED!!"));
   Serial.flush();
   exit(1);
  // Link/associate the LED BUILT digital pin (normally pin 13)
  // with the switch we have just installed/created
 // so that every time the switch is activated the built in
  // LED will be automatically be flipped. Start with LED at HIGH setting.
 int link result = my switches.link switch to output(
                      switch id,
                      LED BUILTIN,
                      HIGH);
 if (link result == link failure) {
    // Linking failed, invalid switch id
    Serial.begin(9600);
    Serial.println(F("Failure to link an output to a switch"));
    Serial.println(F("!!PROGRAM TERMINATED!!"));
    Serial.flush();
    exit(2);
  }
void loop() {
 do {
    if (my switches.read switch(switch id) == switched) {
      // just keep reading, LED BUILTIN will automatically be flipped for us
      // so we dont need to do anything else
  } while (true);
```

Example 3.1 - Turning Multiple LEDs On/Off With Multiple Button & Toggle Switches, Directly Coded

In this example we build on previous examples and see how we can implement and manage a number of button and toggle switches with ease by defining our switch and associated switching output (LED) parameters in an orderly way – we shall use a struct(ure) data type to keep everything we need together.

For the purposes of this example we shall connect two button and two toggle switches, each connected with each type of circuit.

The switch mappings and outputs are:

F	Project Na	me: 3.	1 mixed	switch	nes, no	linking		Date: 4 March 2021
	Swi	tch Confi	gs		Linked Outputs			
Pin	Switch	Туре	Circuit	Туре	Pin	Initial Value		Notes
FIII	Button	Toggle	C1	C2	FIII	LOW	HIGH	
2	X		Χ					No linked output, button will flip in-built
2	^		^					LED by direct coding
3	Х			Х				No linked output, button will flip in-built
3	^			^				LED by direct coding
4		V	V					No linked output, toggle will flip in-built
4		Х	Х					LED by direct coding
_		V		V				No linked output, toggle will flip in-built
5		Х		Х				LED by direct coding

Components required	Circuit schemes
1 x button switch	circuit C1
	Ov (ground) 10k ohm resister Digital read pin +5v pin Arduino Microcontroller
1 v button quitab	Figure 1 – Circuit C1
1 x button switch	CIRCUIT C2 (ground) Digital read pin switch +5v pin Arduino Microcontroller Figure 2 - Circuit c2
1 x toggle switch	CIRCUIT C1 (ground) Digital read pin Arduino Microcontroller Figure 1 – Grout C1
1 x toggle switch	CITCUIT C2 Ov (ground) Digital read pin switch +5v pin Arduino Microcontroller Figure 2 - Circuit c2
2 x 10k ohm resistors	1 each for each circuit C1
4 x LEDs	Standard wiring scheme for
4 x 220 ohm resistors	connected LED

Components required	Circuit schemes
	220 ohm digital pin GND

The sketch can be accessed from github <u>here</u>.

```
Ron D Bentley, Stafford, UK
  Mar 2021
  Example of use of the ez switch lib library
  Example 3.1
  Reading multiple button & toggle switches wired with
  different circuit types with each switch turning associated
  leds on/off.
  This example uses a struct (ure) data type to define the
  switch and led data.
  This example and code is in the public domain and
  may be used without restriction and without warranty.
#include <Arduino.h>
#include <ez_switch_lib.h> // ez_switch_lib .h & .cpp files are stored under
...\Arduino\lararies\ez switch lib\
#define not configured
                       255 // used to indicate if a switch control data entry
has be configured
#define num switches 4
// We will use a struct(ure) data type to keep our switch/LED
// data tidy and readily accessible
byte sw pin;
                         // digital input pin assigned to the switch
 byte sw_circuit_type; // the type of circuit wired to the switch
                         // holds the switch id given by the add.switch function
 byte sw id;
for this switch
                         // digital pin connecting the LED for this switch
 byte sw led pin;
 bool sw_led_status;
                         // current status LOW/HIGH of the LED connected to this
switch
} btl[num switches] = {
                         // 'btl' = buttons, toggles & LEDs
 //.....> <led data, initial setting level>
button_switch, 2, circuit_C1, not_configured, 8, HIGH,
 button_switch, 3, circuit_C2, not_configured, 9,
 toggle switch, 4, circuit C1, not configured, 10, HIGH,
 toggle switch, 5, circuit C2, not configured, 11, LOW
};
// Declare/define the switch instance of given size
Switches my_switches(num_switches);
void setup() {
 // Attach each switch to its defined digital pin/circuit type
  // and store the switch's id back in its struct entry for later use.
 for (byte sw = 0; sw < num switches; sw++) {</pre>
   int switch id = my switches.add switch(
                     btl[sw].sw_type,
                     btl[sw].sw_pin,
                     btl[sw].sw circuit type);
```

```
// Validate the return
    if (switch id < 0) {
      // Error returned - there is a data compatibility mismatch (-2, bad params),
      // or no room left to add switch (-1, add failure).
      Serial.begin (9600);
      Serial.println(F("Failure to add a switch"));
     if (switch id == add failure) {
       Serial.println(F("add switch - no room to create given switch"));
      } else {
        // Can only be that data for switch is invalid
        Serial.println(F("add switch - one or more parameters is/are invalid"));
     Serial.println(F("!!PROGRAM TERMINATED!!"));
      Serial.flush();
     exit(1);
   btl[sw].sw id = switch id; // store given switch id for this sw for use later
   // Now initialise the switch's associated LED and turn on/off according to
preset
    pinMode(btl[sw].sw_led_pin, OUTPUT);
    digitalWrite(btl[sw].sw led pin, btl[sw].sw led status);
}
void loop() {
  // Keep reading the switches we have created and flip their
  // associated LEDs on/off
 do {
    for (byte sw = 0; sw < num switches; sw++) {</pre>
     if (my switches.read switch(btl[sw].sw id) == switched) {
        // Flip between HIGH and LOW each cycle
       btl[sw].sw_led_status = HIGH - btl[sw].sw_led_status;
        digitalWrite(btl[sw].sw led pin, btl[sw].sw led status);
      }
  } while (true);
```

Example 3.2 - Turning Multiple LEDs On/Off With Multiple Button & Toggle Switches, Indirectly Coded

In this example we build on the previous example 3.1 and see how we can implement and manage a number of button and toggle switches <u>without</u> direct coding for the outputs. In the example sketch we remove the direct coding that deals with the switching of the LED outputs and, instead, we use the function <u>link_switch_to_output</u> to associate each switch to an LED output pin. It is essentially the same sketch as in example 3.1, above. Compare this sketch with example 3.1 sketch and note the differences.

For the purposes of this example we shall connect two button and two toggle switches, each connected with each type of circuit.

The switch mappings and outputs are:

F	Project N	ame: 3	.2 mixe	d switc	hes, wi	th linkir	ng	Date: 4 Marcl	h 2021
	Swi	tch Conf	igs		Lir	ked Out	puts		
Pin	Switch	n Type	Circui	t Type	Pin	Initial	Value	Notes	
PIN	Button	Toggle	C1	C2	FIN	LOW	HIGH		
2	Х		Х		8		Х	LED to flip – no direct coding needed the LED	to flip
3	Х			Х	9	Х		LED to flip – no direct coding needed the LED	to flip
4		Х	Х		10		Х	LED to flip – no direct coding needed the LED	to flip
5		Х		Х	11	Х		LED to flip – no direct coding needed the LED	to flip

	1
Components required	Circuit schemes
1 x button switch	circuit C1 ov (ground) Digital read pin 4-5v pin Arduino Microcontroller Figure 1 - Circuit C1
1 x button switch	circuit_C2 Ov (ground) Digital read pin switch +5v pin Arduino Mirocontroller Figure 2 - Circuit C2
1 x toggle switch	Circuit C1 (ground) 10k ohm resister Digital read pin Arduino Microcontroller Figure 1 - Circuit C1
1 x toggle switch	Circuit_C2 Ov (ground) Digital read pin switch +5v pin Arduino Microcontroller Figure 2 - Circuit C2
2 x 10k ohm resistors	1 each for each circuit C1
4 x LEDs	Standard wiring scheme for

Components required	Circuit schemes
4 x 220 ohm resistors	connected LED
	220
	ohm
	digital
	pin ≐ GND
	GND

The sketch can be accessed from github here.

```
Ron D Bentley, Stafford, UK
   Mar 2021
   Example of use of the ez_switch_lib library
   Example 3.2
   Reading multiple button & toggle switches wired with
   different circuit types with each switch linked to an
   output pin using a ez switch lib function, so that when
  activated, the associated switch output pin will be
  automatically flipped, HIGH-LOW etc each time the switch
  is actuated WITHOUT any further coding.
  To demonstrate, the switch associated outputs are connected
   to leds.
   This example and code is in the public domain and
   may be used without restriction and without warranty.
#include <Arduino.h>
#include <ez switch lib.h> // ez switch lib .h & .cpp files are stored under
...\Arduino\libraries\ez switch lib\
#define not_configured
                          255 // used to indicate if a switch control data entry
has be configured
#define num switches 4
// We will use a struct(ure) data type to keep our switch/LED
// data tidy and readily accessible
struct switch control {
                          // struct member meanings:
 byte sw_type;
                            // type of switch connected
 byte sw_pin;
byte sw_circuit_type;
byte sw_id;
                            // digital input pin assigned to the switch
                           // the type of circuit wired to the switch
// holds the switch id given by the add.switch function
for this switch
 byte sw output pin;
                            // digital pin to associate switch to
 bool sw output level;
                            // define the status level of the defined output pin on
set up
                            // 'btl' = buttons, toggles & LEDs
} btl[num switches] = {
 //..... switch data..... <output pin initial setting
level>
 button_switch, 2, circuit_C1, not_configured, 8, HIGH,
button_switch, 3, circuit_C2, not_configured, 9, LOW,
toggle_switch, 4, circuit_C1, not_configured, 10, HIGH,
 toggle_switch, 5, circuit_C2, not_configured, 11, LOW
// Declare/define the switch instance of given size
Switches my_switches(num_switches);
void setup() {
```

```
// Attach each switches to its defined digital pin/circuit type
  // and store the switch's id back in its struct entry for later use.
  for (byte sw = 0; sw < num switches; sw++) {</pre>
   int switch_id = my_switches.add_switch(
                       btl[sw].sw_type,
                       btl[sw].sw pin,
                       btl[sw].sw circuit type);
    // Validate the return
    if (switch id < 0) {
      // Error returned - there is a data compatibility mismatch (-2, bad_params),
      // or no room left to add switch (-1, add failure).
     Serial.begin (9600);
      Serial.println(F("Failure to add a switch"));
      if (switch id == add failure) {
       Serial.println(F("add switch - no room to create given switch"));
      } else {
        // Can only be that data for switch is invalid
        Serial.println(F("add_switch - one or more parameters is/are invalid"));
     Serial.println(F("!!PROGRAM TERMINATED!!"));
      Serial.flush();
      exit(1);
   btl[sw].sw id = switch id; // store given switch id for this sw for use later
    // Now associate the defined out for this switch so that every time the switch
    // is activated the associated output will be automatically be flipped.
    // set the output level to whatever is defined in the initialisation data.
    int link result = my switches.link switch to output(
                        switch id,
                        btl[sw].sw output pin,
                        btl[sw].sw_output_level);
    if (link result == link failure) {
      // Linking failed, invalid switch id
     Serial.begin(9600);
     Serial.println(F("Failure to link an output to a switch"));
     Serial.println(F("!!PROGRAM TERMINATED!!"));
      Serial.flush();
      exit(2);
  }
}
void loop() {
 do {
    for (byte sw = 0; sw < num switches; sw++) {</pre>
      if (my switches.read switch(btl[sw].sw id) == switched) {
       // Just keep reading, the read function will automatically
        // flip the associated switch output pins for us so we
        // dont need to do anything else
  } while (true);
```

Example 4 – Processing More Button & Toggle Switches

In this example we shall build on the previous examples by implementing six switches – three button and three toggle, to show how we are able to keep adding switches of different wiring schemes. This time we shall incorporate outputs to the serial monitor, LEDs and relays, as follows:

The sketch is configured for 6 switches, 3 x toggle and 3 x button with switches performing the following actions:

- toggle 1 switches a relay, without direct coding, using output linking
- toggle 2 switches a led, without direct coding, using output linking
- toggle 3 also switches a led by direct coding, i.e. not via switch linking
- button 1 produces a switch report using a <ez switch lib> function
- button 2 switches a relay, without direct coding, using output linking
- button 3 switches a led, without direct coding, using output linking

The mappings for switches and outputs are:

I	Project N	ame: E	xample	4 - LE	Ds & F	Date: 4 March 2021			
	Swi	tch Confi	gs		Li	nked Out	outs		
Pin	Switch	n Type	Circuit	t Type	Pin	Pin Initial Value		Notes	
PIII	Button	Toggle	C1	C2		LOW	HIGH		
2		Χ	Χ		8		Χ	Relay 1 - no switching coding	
3		Χ		Χ	9	Х		Led 1 - no switching coding	
4		Χ	Χ					Led 3 - needs switching coding	
5	Χ			Χ				Output switch report to serial monitor	
6	Χ		Χ		10		Χ	Relay 2 - no switching coding	
7	Χ			Χ	11	Х		Led 2 - no switching coding	

The switches are polled in succession and processing occurs via a switch-case set of control statements.

We shall also see how we are able to refer to the status of toggle switches outside of them being read by the read switch function and show their status by using a button switch.

To note is that in this example we use a multidimensional array to hold our switch data, rather than a struct(ure) as in example 3.1/3.2 – you decide with approach is best. You will also see that a switch-case series of statements are used to process the switches once triggered.

Make sure to open the serial monitor once the sketch is compiled and uploaded and set to 9600 baud and note the switch circuit schemes are changed from previous set ups, just to further mix things up!

Components required	Circuit schemes
2 x button switch	circuit C2
1 x button switch	Circuit C1 District District

Components required	Circuit schemes
2 x toggle switch	circuit C1 100 conn. 100
1 x toggle switch	circuit C2
3 x 10k ohm resistors	1 each for each circuit_C1
3 x LEDs	Standard wiring scheme
3 x 220 ohm resistors	for connected LED 220 ohm digital pin GND
2 x 5v relays	
Serial monitor	9600 baud

The sketch can be accessed from github <u>here</u>.

```
Ron D Bentley, Stafford, UK
  Mar 2021.
  Example of use of the ez switch lib library
  Reading multiple switches (6) of different types and of mixed wiring
  schemes. Additionally, some switches are linked to a digital
  output such that when they are actuated the linked output level
  is flipped (HIGH->LOW, or LOW->HIGH) automatically without the
  need for any end user coding.
  Switch data in this example is preset in a two dimension array
  and may be varied as appropriate.
  The sketch is configured for 6 switches, 3 toggle and 3 button
  with switches performing the following actions:
  toggle 1 switches a relay, without direct coding, using output linking
  toggle 2 switches a led, without direct coding, using output linking
  toggle 3 also switches a led, but not via switch linking
  button 1 produces a switch report using a ez switch lib function
  button 2 switches a relay, without direct coding, using output linking
  button 3 switches a led, without direct coding, using output linking
  The switches are polled in succession and processing occurs
  via a switch-case set of control statements.
  This example and code is in the public domain and
  may be used without restriction and without warranty.
#include <Arduino.h>
#include <ez switch lib.h> // ez switch lib.h & .cpp files are stored under
...\Arduino\libraries\ez switch lib\
// Declare/define specific 'my data' for 'my project'
#define num switches 6
```

```
// Switch to Pin Macro Definition List:
#define my_toggle_pin_1 2 // digital pin number
#define my_toggle_pin_2 3 // etc
#define my_toggle_pin_3
#define my button pin 1
#define my button pin 2 6
#define my button pin 3
// Digital output pins for linking to switches:
#define relay_1 8 // output pin for relay 1
#define relay_2 9 // output pin for relay 2
#define led_2
#define led 3
                  12 // output pin for led 3
#define not_configured 255 // used to indicate if the my_switch_data switch output
pin is to be configured
// Establish type of switch, assigned digital pin and circuit type
// for each switch we are connecting. Until we present each // switch entry to the add.switch function it will not be
// recorded as configured, hence the use of the final column.
// Array row definitions are:
// [sw][0] = switch type, button or toggle
// [sw][1] = digital input pin for the switch
// [sw][2] = how the switch is wired/connected
// [sw][3] = this stores the switch id returned from add.switch function
//
             to be used in all calls to the library's functions where
             switches are referenced
// [sw][4] = the digital output pin linked to this switch, if defined
// [sw][5] = the level the output pin is to be set to at initialisation (linking),
//
            if a linked output is configured - must be LOW or HIGH
// Note that:
// 'on', 'switched', 'button_switch', 'toggle_switch', 'circuit_C1'
// and 'circuit_C2' are reserved library defined macros.
byte my switch data[num switches][6] =
  // <.....> <.output pin data.>
  toggle_switch, my_toggle_pin_1, circuit_C1, 0, relay_1, HIGH,// linked to relay_1
toggle_switch, my_toggle_pin_2, circuit_C2, 0, led_1, LOW,// linked to led_1
  toggle_switch, my_toggle_pin_3, circuit_C1, 0, not_configured, 0,// not linked,
flip led 3 by direct code
 button switch, my button pin 1, circuit C2, 0, not configured, 0,// not linked,
produces switch report
 button_switch, my_button_pin_2, circuit_C1, 0, relay_2, HIGH,// linked to relay_2
 button_switch, my_button_pin_3, circuit_C2, 0, led 2, LOW // linked to led 2
};
// Declare/define the switch instance of given size
Switches my switches (num switches);
// Set up connected switches as per 'my_switch_data' configs
void setup()
  Serial.begin(9600);
  // Create/install the defined switches...
  create my switches();
  // Set debounce for 20 msecs
  my switches.set debounce(20);
```

```
// initialise the output pin for led 3, as we will deal with
 // flipping this led by direct coding
 pinMode(led 3, OUTPUT);
 digitalWrite(led_3, LOW);
void loop()
 do {
    // Poll all switches - examine each connected switch in turn and, if switched,
    // process its associated purpose.
    for (int sw = 0; sw < num switches; sw++) {</pre>
     byte switch id = my switch data[sw][3]; // extract the switch id for this
      if (my switches.read switch(switch id) == switched) {
       // This switch ('switch id') has been pressed, so process via its switch-
case code
       if (my switches.switches[switch id].switch type == button switch) {
          Serial.print(F("\nbutton switch on digital pin "));
        } else {
          Serial.print(F("\ntoggle switch on digital pin "));
       byte my switch pin = my switches.switches[switch id].switch pin;
        Serial.print(my switch pin);
        Serial.println(F(" triggered"));
        // Move to switch's associated code section
        switch (my switch pin)
          case my toggle pin 1:
            // toggle switch 1 triggers a relay (1) which is a linked output
            // so nothing to do here to process the relay
            Serial.print(F("relay 1 switched"));
            break;
          case my toggle pin 2:
            // toggle switch 2 flips a led (1) which is a linked output
            // so nothing to do here to process the relay
            Serial.print(F("led 1 switched"));
            break;
          case my toggle pin 3:
            // direct coding to flip led 3 following switch actuation (toggle 3)
            static bool led 3 status = LOW; // static because we need to retain
current state between switching
            led 3 status = HIGH - led 3 status; // flip led status
            digitalWrite(led 3, led 3 status);
            Serial.print(F("led 3 switched "));
          case my_button_pin_1:
           // button switch 1 used to reveal the current status of the switch
control structure
            // members, number of free switch control slots and the on/off status
of all
            // all toggle switches as their status is maintained
            my_switches.print_switches(); // confirm we are set up correctly
            // Report number of free switch slots remaining
            Serial.print(F("\nNumber of free switch slots remaining = "));
            Serial.println(my switches.num free switch slots());
            // Report on the current status of the toggle switches
           print_toggle_status();
            Serial.flush();
           break;
          case my button pin 2:
            // button switch 2 triggers a relay (2) which is a linked output
            // so nothing to do here to process the relay
            Serial.print(F("relay 2 switched"));
            break;
          case my button pin 3:
            // button switch 3 flips a led (2) which is a linked output
            // so nothing to do here to process the relay
```

```
Serial.print(F("led 2 switched"));
            break:
          default:
            // Spurious switch index! Should never arise as this is controlled
            // by the for loop within defined upper bound
        Serial.flush(); // flush out the output buffer
      }
  while (true);
// Print the current status/setting of each toggle switch configured.
// We scan down my switch data to pick out toggle switches and if they are
// configured access their status.
void print_toggle_status() {
  Serial.println(F("\nToggle switches setting: "));
  for (byte sw = 0; sw < num switches; sw++)
    if (my switch data[sw][0] == toggle switch) {
      Serial.print(F("toggle switch on digital pin "));
      Serial.print(my_switch_data[sw][1]);
      Serial.print(F(" is "));
      byte switch id = my switch data[sw][3]; // this is the position in the switch
control struct for this switch
      if (my switches.switches[switch id].switch status == on) {
       Serial.println(F("ON"));
      } else {
        Serial.println(F("OFF"));
  }
}
// Create a switch entry for each wired up switch, in accordance
// with 'my' declared switch data.
// add switch params are - switch type, digital pin number and circuit type.
// Return values from add switch are:
      >= 0 the switch control structure entry number ('switch id') for the switch
added,
        -1 no slots available in the switch control structure,
//
        -2 given parameter(s) for switch are not valid.
void create_my_switches() {
  for (int sw = 0; sw < num_switches; sw++) {</pre>
    int switch id =
      my switches.add switch(
        my_switch_data[sw][0], // switch type
        my_switch_data[sw][1], // digital pin number
        my switch data[sw][2]);// circuit type
    if (switch id < 0)
    { // There is a data compatibilty mismatch (-2, bad params),
      // or no room left to add switch (-1, add failure).
      Serial.print(F("Failure to add a switch:\nSwitch entry:"));
      Serial.print(sw);
      Serial.print(F(", data line = "));
      Serial.print(my_switch_data[sw][0]);
      Serial.print(F(", "));
      Serial.print(my_switch_data[sw][1]);
      Serial.print(F(", "));
      Serial.println(my_switch_data[sw][2]);
Serial.println(F("!!PROGRAM TERMINATED!!"));
      Serial.flush();
      exit(1);
```

```
} else {
      // 'switch id' is the switch control slot entry for this switch (sw),
      // so we can use this to know where our switches are
      // in the control structure by keeping a note of them in their
      // my_switch_data config settings.
      my switch data[sw][3] = switch id;
      // Now deal with any linked output requirement
      if (my switch data[sw][4] != not configured) {
        // there is an output defined for this switch, so link it
        int link result =
          my_switches.link_switch_to_output(
                                      // id of switch to link output to
            switch id,
            my_switch_data[sw][4], // digital output pin number
my_switch_data[sw][5]); // initial level, HIGH or LOW
        if (link result == link failure ) {
          // linking failed, invalid switch id
          Serial.println(F("Failure to link an output to a switch"));
          Serial.println(F("!!PROGRAM TERMINATED!!"));
          Serial.flush();
          exit(2);
    }
} // End create my switches
```

Example 5 – Using the Libraries Switch Structure Variables

In this final example we see how we can extend the use of switches by accessing their internal control data and develop a sketch that will allow the hours and minutes of an external timer display to be altered independently of each other.

In this sketch we use one toggle switch to 'activate' the time change cycle, with a linked output to a LED that will automatically illuminate/extinguish on toggle switching, and two button switches, each allowing hours and minutes to be advanced independently.

The code will also allow the button switches to operate in one of two modes – 'single-shot' with each rapid button push advancing the time by +1 (hour or minute) or continuous advance by keeping the button switch pressed.

The external time display is simulated by the sketch with confirmation of <hour:minute> being written to the serial monitor.

The mappings for switches and outputs are:

Project Name: Example 4 – Timer Adjustment Sketch							Date: 4 March 2021			
Switch Configs Linked Outputs						puts				
Pin	Switch Type		Circuit Type		Pin	Initial Value		Notes		
	Button	Toggle	C1	C2		LOW	HIGH			
10			Х	Х	2	Х		Brings adjust code active, LED on when		
10		^						active		
11	Х			Х				Hour adjust button		
12	Х			Х				Minute adjust button		

Components required	Circuit schemes
1 x toggle switch	Circuit_C2 Ov (ground) Digital read pin switch +5v pin Arduino Microcontroller Figure 2 - Circuit C2
2 x button switch	circuit C2 Ov (ground) Digital read pin switch +5v pin Arduin Microcontroller Figure 2 - Circuit C2
1 x LEDs	Standard wiring scheme for
1 x 220 ohm resistors	connected LED 220 ohm digital pin GND

The sketch can be accessed from github <u>here</u>.

```
Ron D Bentley, Stafford, UK
  Mar 2021
  Example of use of the ez switch lib library
  Example 5 - time adjustment sketch
  This example shows the use of button and toggle switches to
  adjust an external time display.
  To implement this successfully we need to use button switches
  to adjust hours and minutes individually. We implement this as
  follows:
  Toggle switch - when on, this will activate the time
  adjustment process bringing the two button switches active.
  When active,
     1 button switch will advance the hours
     1 button switch will advance minutes.
  In this implementation, we wish to advance the times as follows:
     a. if a button is pressed and released immediately, then
        the hour/minute will be advanced by 1, respectively, for
        each press/release
     b. if a button is pressed and kept pressed, then the hour/minute
        will be continually advanced, respectively, automatically
        until it is released.
  To accomplish a) and b) we use the standard read switch function
  for the button switches, but we examine their transition
  status, as we are not interested in a return of 'switched' or not.
  This is an internal flag that is set to true when a button
  switch is pressed until the time it is released. The specific
  switch flag is 'switches[switch id].switch pending'.
  This is true when in transition (pending), false otherwise.
  By using this flag we are able to utilise the button switches
  in either single 'shot mode' or continuous mode for time advancement.
  This is an additional feature of the capabilities of the
  ez switch lib library, and is one that can be used in many
  similar applications.
  As a final feature, the design links the timer adjust toggle switch
  to a LED such that the LED is illuminated when the timer adjust
  mode is active, ie the toggle switch actuated. This linking is
  configured using the ez switch lib function 'link switch to output'.
  This example and code is in the public domain and
  may be used without restriction and without warranty.
#include <Arduino.h>
#include <ez switch lib.h>
#define adjust led
                             2 // LED illuminated when adjust switch on
#define hour adjust switch 10 // a physical button switch, masquerading as a
toggle switch
#define min adjust switch
                            11 // physical button switch, masquerading as a
toggle switch
                             12 // actual toggle switch
#define adjust switch
int hour id, min id, adjust id; // used to record switch ids when declared to
ez switch lib
int hour
            = 0; // initial hour setting
          = 0; // initial minute setting
int min
int now time = 0; // to decide if there has been a time adjustment change
int prev time = 0; // ditto
```

```
#define sensitivity 250 // msecs - used to provide a short delay between switch
reading during adjustments
Switches my_switches(3); // only 3 switches to be declared
void setup() {
                        // we will use the serial monitor to demonstrate
 Serial.begin(9600);
adjustment process
 // declare the switches we wish to use
 adjust_id = my_switches.add_switch(toggle_switch, adjust_switch, circuit_C2);
  // link adjust switch to LED for auto flipping to show switch is on/off
 my switches.link switch to output (adjust id, adjust led, LOW);
 hour id = my switches.add switch (button switch, hour adjust switch, circuit C2);
 min id = my switches.add switch (button switch, min adjust switch, circuit C2);
void loop() {
 // keep polling the adjust switch and action if on
 do {
   my_switches.read_switch(adjust_id); // establish switch status
   if (my_switches.switches[adjust_id].switch_status == on) {
     adjust time(); // adjust switch is on so process any time adjustments
  } while (true);
// Adjust the hours and minutes settings whilst
// the time adjust switch is on
                   void adjust time() {
 do {
    // While the time adjust switch is set,
   // adjust time according to hour/min switches
   my switches.read switch (hour id);
   if (my switches.switches[hour id].switch pending == true) {
     // hour switch is pressed and in transition
     hour = (hour + 1) % 24;
     now time = hour * 60 + min; // minutes since 00:00 hours
   my switches.read switch (min id);
   if (my switches.switches[min id].switch pending == true) {
     // minute switch is pressed and in transition
     min = (min + 1) % 60;
     now time = hour * 60 + min; // minutes since 00:00 hours
   if (now time != prev time) {
     // Either hour button or minute button, or both,
     // have been pressed, so update any external display
     // here with hours/mins.
     // In the absence of an external display, we use the
     // serial monitor to show te adjustments
     if (hour < 10) {
       Serial.print("0"); // leading 0
     Serial.print(hour);
     Serial.print(":");
     if (min < 10) {
       Serial.print("0"); // leading 0
     Serial.println(min);
     prev time = now time;
     delay(sensitivity); // wait a short time between switch presses
   my_switches.read_switch (adjust_id); // establish current adjust switch status
  } while (my switches.switches[adjust id].switch status == on ); // keep going
until deselected
```

Corollary

The <switch_lib> library's functions allow switches of different types, wired in different wiring schemes to be simply read. At their native level they return if a switch has 'switched' or '!switched'. This simple binary result is okay for many applications and uses, however, there are occasions when a little more sophistication and flexibility may be needed. Example 5, above, illustrates how it is possible to make use the library's switch data struct(ure) elements to design advantage; but there are other possibilities!

This section of the User Guide explores some of the deeper <ez_switch_lib> capabilities available to the end user developer.

Switch Mismatching

We add switches into the library's active switch control structure using the add_switch function. This allows us to specify a switch type (button_switch or toggle_switch), the digital pin associated with the switch and how the switch is wired (circuit_C1 or circuit_C2). Ordinarily, we will correctly match the physical switch with the switch type we declare using add_switch.

However, if we mismatch the physical switch type with the declared switch type we can use this to our advantage.

Recap the normal operation of both type of switches:

- button switch it is considered to have been switched when it goes through the cycle OFF-ON-OFF. That is, if we press and release the switch we will read one switching event. At rest it will be OFF
- toggle switch it is considered to have been switched when it goes through either OFF-ON or ON-OFF. That is, if we flip the switch up and down we will read two switching events. At rest it will be either OFF or ON.

Buttons as a Toggles

Let's look at using a button switch masquerading as a toggle switch. That is, we physically connect a button switch but declare it using the add_switch function as a toggle switch. What is the result?

The switch is initially OFF. We now press it and release it. As far as the library is concerned this is a toggle switch that has just been flipped up and down (i.e. switched to ON and then to OFF). We therefore get <u>two switched</u> events, one for the ON event and one for the OFF event. No surprises there.

So, how can this be useful? Well, have a look at the 'Buttons & Lights' game on github (<u>link</u>). This game uses four button switches each associated/linked to a different coloured LED. The objective of the game is to re-enter a random sequence of lights in the correct order using the button switches.

What we want to achieve is for each button press to illuminate a linked LED for the duration of the press only (i.e. for the LED to be turned on when the button is pressed and turned off when released) and for that button guess to be recorded only <u>after</u> the button's release.

We achieve the game's central requirement by:

- 1. using simple button switches
- 2. declaring the button switches as toggle switches using the add_switch function. That is, the button switches will be masquerading as toggle switches to our advantage
- 3. linking each switch to an output pin which has a different coloured LED wired in

The key part of the game, after the switches and associated linkages are made in setup () is:

```
1. for (int sw = 0; sw < num switches; sw++) {
2. byte sw_id = pseudo_toggles[sw][2]; // switch id given by add_switch
3. bool sw_status = my_switches.read_switch(sw_id);
4. if (sw_status == switched &&
5. my_switches.switches[sw_id].switch_status == !on)
6. {
7.    // this switch was pressed on and now switched to off and the
8.    // linked output will have been set to LOW (i.e. LED is off),
9.    // so record it - add to guess list
10.    guesses[0]++;
11.    guesses[guesses[0]] = sw; // record this switch's index
12. }
13. }</pre>
```

Line 3 reads the status of the current switch under consideration. Because we have linked our switches to outputs wired with LEDs, these LEDs will automatically turn to on/off when the button switches are pressed/released, respectively, thereby providing a visual confirmation of switch selection – just as we require, but note that we have not explicitly coded this.

Lines 4 and 5 test if the switch has been switched, but <u>also</u> that the switch has gone from on to off (!on). This is the condition we need to register a single user guess following a complete button press/release cycle. Waiting for the !on condition ensures that the switch's linked out LED is turned to off after button switch release. If we tested for just on, the LED would not be extinguished.

We can only do this because the switch is declared as a toggle switch and its status is therefore always maintained by the library in the switch control struct(ure) whenever actuated - switches [sw id].switch status.

We could not have used button switches declared as button switches to readily achieve the above without a degree of additional direct coding, or similarly toggle switches. What the above example demonstrates is that <ez_switch_lib> can provide a deeper degree of flexibility and capabilities to the developer.

Many Switches, One Interrupt Service Routine (ISR)

The ability to link a digital output pin to a switch so that it will be automatically flipped on switch actuation can be a useful feature to the end user developer. By way of an academic exercise the sketch below was developed to explore the use of this feature to link multiple switches (of different types and wiring schemes) to a single interrupt service routine or ISR.

Recall that the linking of a switch to an output pin¹ for automatic switching/flipping only occurs via use of the switch read function read_switch. The other two switch reading functions, read_button_switch and read_toggle_switch, will not process any linked outputs. This feature provides a degree of flexibility in that there may be circumstances when a switch needs to be read without affecting any linked output.

The sketch does not include very much code but it is extensively documented and the reader should be well versed at this point with the approach adopted. The switch mappings and outputs for this sketch are:

F	Project Name: Corollary – Multiple Switches, One ISR								Date:	29 March 2021
	Switch Configs					nked Out	outs			
Pin	Switch Type		Circuit Type		Pin	Initial Value		Notes		
	Button	Toggle	C1	C2	FIII	LOW	HIGH			
2								Interrupt pin, not physical connected		
3	Χ		Χ		2	Х		1 st button switch		
4	Χ		X		2	Χ		2 nd button swich		
5		Χ	X		2	Χ		1 st toggle switch		
6		Χ		Χ	2	Х		2 nd toggle swich		

The sketch may be accessed from github <u>here</u>.

```
// Ron Bentley, Stafford UK
// March 2021
11
\ensuremath{//} This example and code is in the public domain and may be used without
// restriction and without warranty.
11
         Example sketch - Multiple switches handled by a single interrupt
          11
// This sketch demonstrates how the ez switch lib may be used to handle multiple
// switches (button & toggle switches in this example) with a single interrupt
routine.
//
// The use of the ez switch lib library for switches provides:
//
   * switch type independence
    * switch circuit type independence
   * automatic multiple switch debounce handling
//
    ^{\star} parallel switching capabilities, and
    * automatic interrupt handling for all switches
// The sketch is designed such that when a toggle switch is switched to the 'on'
// position, or a button switch is pressed AND released a linked output connected
// to a common interrupt pin will cause the associated interrupt handler to be
// fired to process the switch 'on' event.
// NB, and to recap:
// 1. a toggle switch will fire the interrupt when set to 'on'. Setting it back
//
      off does not fire the interrupt
//
    2. a button switch will ONLY fire the interrupt when pressed 'on' AND
//
       then released
11
       to off. The interrupt fires on completion of the button switch cycle.
11
// Note that:
  1. error checking on switch set ups has been removed post development.
    2. the serial monitor is used to confirm the correct operation of the sketch.
// The sketch will use digital pin 2 as the common interrupt pin and
// pins 3, 4, 5 and 6 as the switch pins.
```

We use the function <u>link switch to output</u> to achieve this capability.

```
// For an understanding of the capabilities of the 'ez switch lib' library see
// the USER GUIDE:
// https://github.com/ronbentley1/ez_switch_lib-Arduino-
Library/blob/main/ez switch lib user guide%2C%20v1.02.pdf
#include <ez switch lib.h>
       interrupt pin = 2; // external interrupt pin
#define num switches
// 'my switches' layout.
// one row of data for each switch to be configured, as follows:
// [][0] = switch type
// [][1] = digital output pin connected to switch
// [][2] = the switch_id provided by the add_switch function for the
          switch declared
// [][3] = the circuit type connecting the switch, here the first 3 switches
//
          will have 10k ohm pull down resistors wired, whilst the 4th switch
//
          will be wired directly without an external pull down resistor
byte my_switches[num_switches][4] =
 button_switch, 3, 0, circuit_C1,
button_switch, 4, 0, circuit_C1,
toggle_switch, 5, 0, circuit_C1,
 toggle switch, 6, 0, circuit C2
// Create the 'Switches' instance (ms) for the given number of switches
Switches ms (num switches);
void setup() {
  // Add all switches to library switch control structure
  // and link all to same interrupt pin as a linked output
  for (byte sw = 0; sw < num_switches; sw++) {</pre>
   my_switches[sw][2] = ms.add_switch(
                        my switches[sw][0], // switch type
                        my switches[sw][1], // digital pin switch is wired to
                        my switches[sw][3]); // type of circuit switch is wired as
   ms.link switch to output(
    my_switches[sw][2], // switch id
                       // digital pin to link to for interrupt
    interrupt_pin,
    LOW);// start with interrupt pin LOW, as interrupt will be triggered on RISING
  // Now establish the common interrupt service routine (ISR) that
  // will be used for all declared switches
 attachInterrupt(
   digitalPinToInterrupt(interrupt_pin),
    switch_ISR, // name of the sketch's ISR function to handle switch interrupts
    RISING); // trigger on a rising pin value
  Serial.begin (115200);
} // end of setup function
void loop() {
  // Keep testing switch, and let the interrupt handler do its thing
  // once a switch is switched to 'on'
 for (byte sw = 0; sw < num switches; sw++) {
   ms.read switch(my switches[sw][2]); // my switches[sw][2] is the switch id
                                       // for switch sw
  }
}
\ensuremath{//} ISR for handling interrupt triggers arising from associated switches
// when they transition to on. The routine knows which switch has generated
// the interrupt because the ez switch lib switch read functions record the
// actuated switch in the library variable 'last_switched_id'.
```

```
// The routine does nothing more than demonstrate the effectiveness of the
// use of a single ISR handling multiple switches by using the serial monitor
// to confirm correct operation.
void switch ISR()
 // Reset the interrupt pin to LOW, so that any other switch will fire the
 // interrupt whist one or more switches in transition stage
 byte switch_id = ms.last_switched_id; // switch id of switch currently
                                      // switched to on
 digitalWrite(ms.switches[switch_id].switch_out_pin, LOW);
 // For button switches only, reset the linked output pin status to LOW so that
  // it will trigger the interrupt at every press/release cycle.
 if (ms.switches[switch_id].switch_type == button_switch) {
   ms.switches[switch id].switch out pin status = LOW;
 Serial.print("** Interrupt triggered for switch id ");
 Serial.println(switch id); // 'this switch id' is the id of the triggering switch
 Serial.flush();
} // end of switch ISR
```

There are other options to be explored.....enjoy!